NAG Fortran Library Routine Document

G13BDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13BDF calculates preliminary estimates of the parameters of a transfer function model.

2 Specification

```
SUBROUTINE G13BDF(R0, R, NL, NNA, S, NWDS, WA, IWA, WDS, ISF, IFAIL)
INTEGER           NL, NNA(3), NWDS, IWA, ISF(2), IFAIL
real              R0, R(NL), S, WA(IWA), WDS(NWDS)
```

3 Description

The routine calculates estimates of parameters $\delta_1, \delta_2, \ldots, \delta_p$, $\omega_0, \omega_1, \ldots, \omega_q$ in the transfer function model

$$y_t = \delta_1 y_{t-1} + \delta_2 y_{t-2} + \ldots + \delta_p y_{t-p} + \omega_0 x_{t-b} - \omega_1 x_{t-b-1} - \ldots - \omega_q x_{t-b-q}$$

given cross correlations between the series x_t and lagged values of y_t:

$$r_{xy}(l), \quad l = 0, 1, \ldots, L$$

and the ratio of standard deviations s_y/s_x, as supplied by G13BCF.

It is assumed that the series x_t used to calculate the cross correlations is a sample from a time series with true autocorrelations of zero. Otherwise the cross correlations between the series b_t and a_t, as defined in the description of G13BAF, should be used in place of those between y_t and x_t.

The estimates are obtained by solving for $\delta_1, \delta_2, \ldots, \delta_p$ the equations

$$r_{xy}(b+q+j) = \delta_1 r_{xy}(b+q+j-1) + \cdots + \delta_p r_{xy}(b+q+j-p), \quad j = 1, 2, \ldots, p$$

then calculating

$$\omega_i = \pm(s_y/s_x) r_{xy}(b+i) - \delta_1 r_{xy}(b+i-1) - \cdots - \delta_p r_{xy}(b+i-p), \quad i = 0, 1, \ldots, q$$

where the '$+$' is used for ω_0 and '$-$' for ω_i, $i > 0$.

Any value of $r_{xy}(l)$ arising in these equations for $l < b$ is taken as zero. The parameters $\delta_1, \delta_2, \ldots, \delta_p$ are checked as to whether they satisfy the stability criterion.

4 References

Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* (Revised Edition) Holden-Day

5 Parameters

1: R0 – ***real*** *Input*

On entry: the cross correlation between the two series at lag 0, $r_{xy}(0)$.

Constraint: $-1.0 \leq$ R0 ≤ 1.0.

2: R(NL) – ***real*** array *Input*

On entry: the cross correlations between the two series at lags 1 to L, $r_{xy}(l)$, for $l = 1, 2, \ldots, L$.

Constraint: $-1.0 \leq \mathrm{R}(i) \leq 1.0$, for $i = 1, 2, \ldots, \mathrm{NL}$.

3: NL – INTEGER *Input*

On entry: the number of lagged cross correlations, L, in the array R.

Constraint: $\mathrm{NL} \geq \max(\mathrm{NNA}(1) + \mathrm{NNA}(2) + \mathrm{NNA}(3), 1)$.

4: NNA(3) – INTEGER array *Input*

On entry: the transfer function model orders in the standard form b, q, p (i.e., delay time, number of moving-average MA-like followed by number of autoregressive AR-like parameters).

Constraint: $\mathrm{NNA}(i) \geq 0$, for $i = 1, 2, 3$.

5: S – ***real*** *Input*

On entry: the ratio of the standard deviation of the y series to that of the x series, s_y/s_x.

Constraint: $\mathrm{S} > 0.0$.

6: NWDS – INTEGER *Input*

On entry: the exact number of parameters in the transfer function model.

Constraint: $\mathrm{NWDS} = \mathrm{NNA}(2) + \mathrm{NNA}(3) + 1$.

7: WA(IWA) – ***real*** array *Workspace*

8: IWA – INTEGER *Input*

On entry: the dimension of the array WA as declared in the (sub)program from which G13BDF is called.

Constraint: $\mathrm{IWA} \geq \mathrm{NNA}(3) \times (\mathrm{NNA}(3) + 1)$.

9: WDS(NWDS) – ***real*** array *Output*

On exit: the preliminary estimates of the parameters of the transfer function model in the order of $q + 1$ MA-like parameters followed by the p AR-like parameters. If the estimation of either type of parameter fails then these parameters are set to 0.0.

10: ISF(2) – INTEGER array *Output*

On exit: indicators of the success of the estimation of MA-like and AR-like parameters respectively. A value 0 indicates that there are no parameters of that type to be estimated. A value of 1 or -1 indicates that there are parameters of that type in the model and the estimation of that type has been successful or unsuccessful respectively. Note that there is always at least one MA-like parameter in the model.

11: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, NNA$(i) < 0$, for $i = 1, 2, 3$,
or NL $< \max(\text{NNA}(1) + \text{NNA}(2) + \text{NNA}(3), 1)$,
or R0 < -1.0 or R0 > 1.0,
or R$(i) < -1.0$ or R$(i) > 1.0$, for some $i = 1, 2, \ldots, \text{NL}$,
or S ≤ 0.0,
or NWDS $\neq \text{NNA}(2) + \text{NNA}(3) + 1$,
or IWA $< \text{NNA}(3) \times (\text{NNA}(3) + 1)$.

7 Accuracy

Equations used in the computations may become unstable, in which case results are reset to zero with array ISF values set accordingly.

8 Further Comments

The time taken by the routine is roughly proportional to NWDS^3.

9 Example

The example program reads the cross correlations between 2 series at lags 0 to 6. It then reads a (3,2,1) transfer function model and calculates and prints the preliminary estimates of the parameters of the model.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G13BDF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NLMAX, NWDSMX, IWAMAX
      PARAMETER        (NLMAX=10,NWDSMX=5,IWAMAX=20)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             R0, S
      INTEGER          I, IFAIL, IWA, NL, NWDS
*     .. Local Arrays ..
      real             R(NLMAX), WA(IWAMAX), WDS(NWDSMX)
      INTEGER          ISF(2), NNA(3)
*     .. External Subroutines ..
      EXTERNAL         G13BDF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G13BDF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NL
      READ (NIN,*) R0
      IF (NL.GT.0 .AND. NL.LE.NLMAX) THEN
         READ (NIN,*) (R(I),I=1,NL)
         READ (NIN,*) (NNA(I),I=1,3)
         READ (NIN,*) S
         NWDS = NNA(2) + NNA(3) + 1
         IWA = NNA(3)*(NNA(3)+1)
         IF (NWDS.LE.NWDSMX .AND. IWA.LE.IWAMAX) THEN
```

```
            IFAIL = 0
*
            CALL G13BDF(R0,R,NL,NNA,S,NWDS,WA,IWA,WDS,ISF,IFAIL)
*
            WRITE (NOUT,*)
            WRITE (NOUT,99999) 'Success/failure indicator', ISF(1),
     +        ISF(2)
            WRITE (NOUT,*)
            WRITE (NOUT,99999) 'Transfer function model B, Q, P =',
     +        (NNA(I),I=1,3)
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Parameter initial estimates'
            WRITE (NOUT,99998) (WDS(I),I=1,NWDS)
          END IF
        END IF
        STOP
*
99999 FORMAT (1X,A,3I4)
99998 FORMAT (1X,4F10.4)
      END
```

9.2 Program Data

```
G13BDF Example Program Data
     6
 -0.0155
  0.0339 -0.0374 -0.2895 -0.3430 -0.4518 -0.2787
     3     2     1
  1.9256
```

9.3 Program Results

```
 G13BDF Example Program Results

 Success/failure indicator   1   1

 Transfer function model B, Q, P =   3   2   1

 Parameter initial estimates
    -0.5575    0.3166    0.4626    0.6169
```